2 Modelització amb GLPK. Variables enteres i binàries, variables indexades per dos conjunts

En aquesta pràctica continuarem aprenent a fer funcionar el paquet GLPK i l'utilitzarem per resoldre problemes d'enunciat. El llenguatge del GLPK, anomenat GNUMathProg o GMPL(essencialment llenguatge AMPL), ens ajudarà no només a resoldre els problemes sinó també a plantejar-los.

2.1 Un exemple amb variables enteres, indexades en dos conjunts (matrius de variables)

Una empresa dedicada a la fabricació de components d'ordinador té dues fàbriques que produeixen, respectivament, 800 i 1500 peces mensuals. Aquestes peces han de ser transportades a tres botigues que necessiten 1000, 700 i 600 peces, respectivament. Les despeses de transport, en cèntims d'euro per peça, són les de la taula següent:

	botiga 1	botiga 2	botiga 3
fàbrica A	30	70	10
fàbrica B	20	20	60

Com ha d'organitzar-se el tranport per minimitzar el cost?

Modelització/planteig:

Hem de decidir quines quantitats hem de transportar des de cada una de les fàbriques a cada una de les botigues.

Tenim 2×3 variables enteres no negatives, que podem indexar usant dos índexs, el primer indica la fàbrica i el segon, la botiga

$x_{A,1}$ = nombre de peces des de fàbrica A a botiga 1
$x_{A,2}$ = nombre de peces des de fàbrica A a botiga 2
$x_{A,3}$ = nombre de peces des de fàbrica A a botiga 3

$x_{B,1}$ = nombre de peces des de fàbrica B a botiga 1
$x_{B,2}$ = nombre de peces des de fàbrica B a botiga 2
$x_{B,3}$ = nombre de peces des de fàbrica B a botiga 3

La funció objectiu és el cost, que volem minimitzar:

$$z = x_{A,1}\, c_{A,1} + x_{A,2}\, c_{A,2} + x_{A,3}\, c_{A,3} + x_{B,1}\, c_{B,1} + x_{B,2}\, c_{B,2} + x_{B,3}\, c_{B,3},$$

on $c_{i,j} =$ cost de transportar una peça de la fàbrica i a la botiga j.

Les restriccions:

capacitat de producció

$$\left.\begin{array}{ll}\text{fàbrica A} & x_{A,1}+x_{A,2}+x_{A,3}\le 800\\ \text{fàbrica B} & x_{B,1}+x_{B,2}+x_{B,3}\le 1500\end{array}\right\}\ \text{que podríem resumir en}$$

$$x_{i,1}+x_{i,2}+x_{i,3}\le \text{capacitat fàbrica } i, \quad \text{per a } i=A,B$$

demanda

$$\left.\begin{array}{ll}\text{botiga 1} & x_{A,1}+x_{B,1}\ge 1000\\ \text{botiga 2} & x_{A,2}+x_{B,2}\ge 700\\ \text{botiga 3} & x_{A,3}+x_{B,3}\ge 600\end{array}\right\}\ \text{que podríem resumir en}$$

$$x_{A,j}+x_{B,j}\ge \text{demanda botiga } j, \quad \text{per a } j=1,2,3$$

Programa escrit en GNU MathProg

```
# Comencem declarant els dos conjunts d'índexs que utilitzarem

set Fabriques;
set Botigues;

# Declarem el paràmetre-vector de capacitat de producció
#                    indexat al conjunt Fabriques

param capacitat_produccio{Fabriques};

# Declarem el paràmetre-vector de demanda
#                    indexat al conjunt Botigues
param demanda{Botigues};

# Declarem el paràmetre-matriu de cost de transport
#          indexat a Fabriques (files) i Botigues (columnes)
param cost_transport{Fabriques,Botigues};

# decl. variable matriu indexada a Fabriques per Botiques
var x{Fabriques,Botigues}>=0,integer;

minimize cost_total: sum{i in Fabriques, j in Botigues} x[i,j]*
    cost_transport[i,j];

# Restriccions indexades
s.t. r_capacitat{i in Fabriques}: sum{j in Botigues} x[i,j] <=
    capacitat_produccio[i];
s.t. r_demanda{j in Botigues}: sum{i in Fabriques} x[i,j] >= demanda[j];

data;
# especifiquem els elements dels conjunts
set Fabriques:= A B;
set Botigues:= 1 2 3;

# Paràmetre matricial
param cost_transport:
    1     2     3:=
A   30    70    10
B   20    20    60;

# Dos paràmetres vectorials:
```

```
param capacitat_produccio:=
A 800 B 1500;

param demanda:=
1 1000 2 700 3 600;


end;
```

Fixem-nos en

- Hem hagut de declarar dos conjunts d'índexs, `Fabriques` i `Botigues`.
- Hi ha paràmetres–vector indexats en un dels dos conjunt d'índex: `demanda{Botigues}` i `capacitat_produccio{Fabriques}`.
- Tenim el paràmetre–matriu `cost_transport` indexat a Fabriques (primer índex) per Botigues (segon índex).
- Tenim una matriu de variables, indexades als conjunts Fabriques (primer índex) per Botiques (segon índex).
- En el bloc de dades introduïm el paràmetre–matriu `cost_transport` utilitzant la sintàxi específica que hem explicat a l'exemple anterior.

2.2 Opció de sortida: escriure el problema detallat en un arxiu, tal com glpsol l'ha interpretat

En l'exemple anterior hem indexat les restriccions. Si bé en realitat són 5 (dues per a la capacitat de producció i tres per a la demanda) les hem escrit en tan sols dues línies:

`s.t. r_capacitat{i in Fabriques}: sum{j in Botigues} x[i,j] <= capacitat_produccio[i];`
(en realitat són dues restriccions, una per a $i = A$ i l'altra per a $i = B$)

`s.t. r_demanda{j in Botigues}: sum{i in Fabriques} x[i,j] >= demanda[j];`
(en realitat són tres restriccions, una per a cada botiga ($j = 1, 2, 3$)).

L'opció `--wlp nom_arxiu.prg` afegida quan executem glpsol ens produeix un arxiu amb el programa tal com l'ha entès el glpsol (totes les restriccions). [1] Per exemple, si l'arxiu `transport.mod` conté el model de l'exemple anterior, la comanda

```
glpsol -m transport.mod -o transport.sol --wlp transport.prg
```

genera dos arxius, el de solucions `transport.sol` i l'arxiu `transport.prg` següent:

...

[1] En algunes versions de glpk se substitueix l'opció `--wlp` per una altra, consulteu l'ajuda amb `glpsol help` si no funciona.

```
Subject To
r_capacitat(A): + x(A,1) + x(A,2) + x(A,3) <= 800
r_capacitat(B): + x(B,1) + x(B,2) + x(B,3) <= 1500
r_demanda(1): + x(A,1) + x(B,1) >= 1000
r_demanda(2): + x(A,2) + x(B,2) >= 700
r_demanda(3): + x(A,3) + x(B,3) >= 600
```

Resulta molt útil aquesta opció, ja que ens permet detectar errors en el codi del model (és a dir, els índexos eren els correctes? ens hem deixat alguna restricció? les hem escrites correctament? les dades no contenien cap error? ...).

2.3 Un exemple de variables binàries

Hem d'assignar les tasques T1, T2, T3 i T4 a quatre persones P1, P2, P3 i P4, donant a cada persona una tasca i de manera que la qualificació global sigui màxima, on cada persona té una qualificació diferent per fer cada tasca, donada per la taula

	T_1	T_2	T_3	T_4
P_1	10	3	4	1
P_2	3	6	4	9
P_3	1	7	4	3
P_4	3	2	1	5

(1)

Recordem que per resoldre aquest problema d'assignació es defineix una matriu 4×4 de variables binàries:

$$x_{ij} = \begin{cases} 1 & \text{si s'assigna a la persona } i \text{ la tasca } j \\ 0 & \text{en altre cas} \end{cases}$$

La restricció que la persona i tingui assignada exactament una tasca, traduïda a una condició sobre les variables binàries, és

$$x_{i1} + x_{i2} + x_{i3} + x_{i4} = 1$$

i la restricció que la tasca j sigui assignada a una sola persona és

$$x_{1j} + x_{2j} + x_{3j} + x_{4j} = 1$$

Finalment, l'objectiu és maximitzar la suma de qualificacions de les assignacions, és a dir maximitzar la funció

$$\sum_{i=1}^{4} \sum_{j=1}^{4} q_{ij}\, x_{ij}$$

on q_{ij} és la qualificació de la persona i per a fer la tasca j donada a la taula (**??**).

El codi glpk per resoldre aquest problema seria molt senzill:

```
set persones;
set tasques;

param qualif{persones,tasques};

```

```
6 var x{persones,tasques} binary;
7
8 maximize z:sum{i in persones,j in tasques} qualif[i,j] * x[i,j];
9
10 subject to
11 TascaTePersona{j in tasques}: sum{i in persones} x[i,j] =1;
12 PersonaTeTasca{i in persones}: sum{j in tasques} x[i,j] =1;
13
14 data;
15 set persones:= P1 P2 P3 P4;
16 set tasques:= T1 T2 T3 T4;
17
18 param qualif:
19    T1 T2 T3 T4:=
20 P1  10 3  4  1
21 P2  3  6  4  9
22 P3  1  7  4  3
23 P4  3  2  1  5;
24 end;
```

Observem només com declarem les variables binàries a la línia 6 i com les vuit restriccions es poden escriure en dues línies (11 i 12), indexant a tasques i a persones, respectivament. Es pot comprovar amb l'opció `--wlp` que en realitat són 8 restriccions.

2.4 Exercicis

1. La taula següent mostra la informació corresponent a la utilitat (en euros) i al pes (en kg) d'una sèrie d'objectes que volem enviar en una panera de Nadal a un client.

objecte	O_1	O_2	O_3	O_4	O_5	O_6	O_7	O_8	O_9	O_{10}
pes	55	50	40	35	30	30	15	15	10	5
utilitat	100	80	75	70	60	55	25	20	20	15

   Quins d'aquests objectes hem de seleccionar a la panera per tal de maximitzar la utilitat, si el pes màxim permès per una companyia de transport de mercaderies és de 100 kg?

2. El responsable d'educació d'una mancomunitat de municipis està organitzant la distribució dels estudiants que el curs vinent començaran ESO en els tres instituts existents. És conscient que els alumnes que siguin assignats a un centre fora de la seva zona hauran de ser traslladats en autocar a compte de la mancomunitat, i vol minimitzar el nombre total de kilòmetres fets en autocar pels estudiants. La mancomunitat està dividida en cinc sectors (A, B, C, D, E) i els instituts estan situats als sectors B, C i E. En la taula següent es recull el nombre d'estudiants de cada sector que cal repartir en els tres instituts i la distància en kilòmetres des de cada sector a cada institut:

	distància a l'institut			
sector	institut situat al sector B	institut situat al sector C	institut situat al sector E	nombre d'estudiants
A	5	8	6	70
B	0	4	12	50
C	4	0	7	10
D	7	2	5	80
E	12	7	0	40
total				250

Cada institut té capacitat per a 90 estudiants nous.

(a) Com s'ha de fer l'assignació d'estudiants als instituts minimitzant la suma del nombre de kilòmetres fets per tots els estudiants?

(b) Suposem que, a més de les tres escoles esmentades, hi ha la possibilitat d'oferir 1er d'ESO en un institut situat al sector D. S'oferirien 40 places només a estudiants que viuen en aquest sector i suposaria un cost anual fix de 10.000 euros.

A canvi, s'estalviaria el cost de transport en autocar d'aquests estudiants. Es considera que el cost de traslladar en autocar un estudiant és de 100 euros per kilòmetre i any. Decidiu si val la pena o no reobrir l'institut del sector D.

3. Al campionat europeu de futbol s'han de jugar aquesta setmana 5 partits, als quals s'hi ha d'enviar un equip arbitral. Les despeses de transport (en milers d'euros) de cada un dels equips disponibles als partits a jugar es recullen en la taula següent:

equips arbitrals	A	B	C	D	E
partit 1	5	6	40	43	28
partit 2	50	51	8	6	32
partit 3	60	55	50	30	25
partit 4	50	45	40	40	20
partit 5	30	31	37	32	25

Com s'ha de fer l'assignació d'equip arbitral a cada partit per tal de minimitzar costos de transport.

4. Una fusteria produeix taulons de fusta de quatre mides: petites, mitjanes, gruixudes i extra-gruixudes. Els taulons es poden fabricar en qualsevol dels tres tipus de màquines següents: bàsica, normal i especial. Les capacitats de processament de cada màquina (en centímetres de tauló per hora) són les que figuren a la taula:

	tipus de màquina		
mida tauló	bàsica	normal	especial
petita	350	650	850
mitjana	300	450	750
gruixuda	250	400	650
extragruixuda	150	250	350

Suposem que cada màquina es pot usar fins a 300 hores al mes i que els costos de processament per hora per a cada tipus de màquina són 25 euros per a la bàsica, 40 per a la normal i 60 per a l'especial.

Com es pot satisfer una demanda mensual de 12000 cm de tauló de mida petita, 9000 cm de mida mitjana, 8000 de gruixuda i 8000 d'extragruixuda amb cost mínim?

5. Entre els sis districtes d'una ciutat hi ha distribuïts set centres públics d'atenció mèdica. Cadascun dóna servei, exclusivament, als ciutadans residents en alguns dels districtes més propers al centre mèdic. En la taula següent s'indica aquesta distribució de servei dels centres C_i als districtes D_j.

	C_1	C_2	C_3	C_4	C_5	C_6	C_7
D_1	•		•		•		
D_2		•	•			•	
D_3	•			•	•		
D_4			•				•
D_5		•		•	•		•
D_6	•				•	•	

Determineu quin és el mínim nombre de centres que cal mantenir, i quins són, per assegurar l'atenció sanitària a tots els districtes de la ciutat.

6. El cap de producció d'una planta química està intentant trobar un canvi de torns beneficiós. Cada dia està dividit en tres torns de 8 hores (0 a 8, 8 a 16 i 16 a 24), denotats per torn de nit, de matí i de tarda, respectivament. La planta està en funcionament totes les hores i tots els dies i necessita com a mínim els treballadors que apareixen en la taula

	dll	dt	dc	dj	dv	ds	dg
nit	5	3	2	4	3	2	2
matí	7	8	9	5	7	2	5
tarda	9	10	10	7	11	2	2

Els acords amb els sindicats posen condicions a l'organització dels torns:

- Cada treballador s'ha d'assignar a un dels tres torns només (o nit o matí o tarda), durant tot el mes.
- Cada empleat treballa quatre dies consecutius i en descansa tres.

En total l'empresa té 60 treballadors.

Com s'han d'organitzar els torns cada un dels dies de la setmana? Pot reduir la plantilla?

7. Una multinacional té dues fàbriques F_1 i F_2, dedicades a la producció de neveres i rentadores. La capacitat de producció mensual de cadascuna d'elles es dóna a la taula

	fàbrica	
producte	F_1	F_2
neveres	5000	8000
rentadores	7000	4000

La multinacional ven a tres clients C_1, C_2 i C_3, amb demandes mensuals:

	client		
producte	C_1	C_2	C_3
neveres	4000	5000	4000
rentadores	3000	3000	4000

Els articles es distribueixen utilitzant el tren com a mitjà de transport. La taula següent resumeix els costos de transport per unitat des de cada fàbrica a cada client (no hi ha distinció entre els productes)

	client		
fàbrica	C_1	C_2	C_3
F_1	6	14	7
F_2	10	8	15

A més, per problemes d'espai, el número d'unitats que poden enviar-se mensualment des de cada fàbrica a cada client està limitat. La següent taula indica el número màxim d'unitats de producte (neveres o rentadores) que pot enviar-se

	client		
fàbrica	C_1	C_2	C_3
F_1	6000	3000	8000
F_2	3000	9000	3000

Determineu com s'ha de realitzar la distribució des de cada fàbrica fins cada client amb objecte de minimitzar el cos total de transport i satisfer les necessitats dels clients.

8. Una gran companyia tèxtil té dues plantes de producció, dos orígens de materies primeres i tres centres de venta. El cost de transport entre els orígens i les plantes i entre les plantes i els mercats és el següent (en euros per tona):

	planta A	planta B
origen 1	1	1.50
origen 2	2	1.50

	mercat 1	mercat 2	mercat 3
planta A	4	2	1
planta B	3	4	2

Es disposa de 10 tones de l'origen 1 i de 15 tones de l'origen 2. Els tres centres de vendes necessiten 8 tones, 14 tones i 3 tones, respectivament. La capacitat de processament de les plantes és il.limitada.

Trobeu la manera de fer l'enviament dels orígens a les plantes i de les plantes als mercats que minimitza el cost total del transport.